

Programming = proving?
The Curry-Howard correspondence today

Ninth lecture

Sisyphus happy:
infinite data types, proofs by coinduction,
and reactive programming

Xavier Leroy

Collège de France

2019-01-16



COLLÈGE
DE FRANCE
— 1530 —

Inductive types, inductive predicates

The lecture of Nov 28th 2018 presented inductive types and inductive predicates, a powerful mechanism to

- define data types and logical predicates
- that are finitely generated by constructors
- and used by structural induction and case analysis.

Example: the natural numbers in Coq.

```
Inductive nat : Type := 0: nat | S: nat -> nat.
```

```
Fixpoint add (n m: nat) {struct n} : nat :=  
  match n with 0 => m | S n' => S (add n' m) end.
```

```
Inductive even : nat -> Prop :=  
  | even_0: even 0  
  | even_S: forall n, even n -> even (S (S n)).
```

Infinite data

How can we declare and work with **infinite data structures**?

For instance:


- Streams, that is, infinite lists.
A stream = a pair of a value and of a stream.
- Infinite binary trees.
An infinite binary tree = a triple (left subtree, value, right subtree).

Note: “infinite” here means “potentially infinite”: a terminating program will only traverse a finite part of the structure.

Computer representation of infinite data

1- By directed graphs with cycles.

Example: a finite deterministic automaton over the alphabet $\{0, 1\}$ encodes an infinite binary tree (with, at each node, a Boolean “accepting / not accepting”).



A finite graph can only represent **regular** infinite structures, that is, those having finitely many different sub-structures.

Regular infinite: the stream $0.1.2.0.1.2.0.1.2. \dots$

Non-regular infinite: the stream of integers $0.1.2.3.4.5.6.7.8.9. \dots$

Computer representation of infinite data

2- By delayed / on-demand evaluation of sub-structures.

Evaluation explicitly delayed by a function:

```
type 'a stream = unit -> 'a cell
and 'a cell = Cons of 'a * 'a stream
let tail s = match s() with Cons(h,t) -> t
```

On-demand evaluation via an explicit lazy type, as in OCaml:

```
type 'a stream = 'a cell Lazy.t
and 'a cell = Cons of 'a * 'a stream
let tail s = match Lazy.force s with Cons(h,t) -> t
```

On-demand evaluation by default, as in Haskell:

```
data Stream a = Cons a (Stream a)    (* implicitly "lazy" *)
```

This lecture

How to model infinite data structures and reason upon them?

- Classic set-theoretic approach: greatest fixed points.
- Proof theoretic approach: infinite trees and infinite derivations.
- Coalgebraic approach: “codata” defined by their observations.

Two applications:

- The partiality monad, to do general recursion in type theory.
- Reactive programming, viewed as programming over infinite streams, or maybe not...

|

Greatest fixed points

Least fixed point, greatest fixed point

Let A be a set and $F : \mathcal{P}(A) \rightarrow \mathcal{P}(A)$ a monotonically increasing function:
if $X \subseteq Y$ then $F(X) \subseteq F(Y)$.

If $F(X) \subseteq X$ we say that X is stable by F .

If $X \subseteq F(X)$ we say that X is consistent for F .

Theorem (Knaster, Tarski, Kleene)

The set $\{x \mid x = F(x)\}$ of fixed points of F is a complete lattice.

In particular,


$\mu F \stackrel{\text{def}}{=} \bigcap \{X \mid X \text{ stable by } F\}$ is the least fixed point of F

and it is the limit of the increasing sequence $\emptyset, F(\emptyset), F(F(\emptyset)), \dots$

$\nu F \stackrel{\text{def}}{=} \bigcup \{X \mid X \text{ consistent for } F\}$ is the greatest fixed point of F

and it is the limit of the decreasing sequence $A, F(A), F(F(A)), \dots$

Fixed points, graphically



Induction and co-induction

Induction principle: if X is stable by F , then $\mu F \subseteq X$.

In other words: to show $\forall a \in \mu F, a \in X$,
it suffices to show $\forall a \in F(X), a \in X$.

Coinduction principle: if X is consistent for F , then $X \subseteq \nu F$.

In other words: to show $\forall a \in X, a \in \nu F$,
it suffices to show $\forall a \in X, a \in F(X)$.

Alternative: to show $a \in \nu F$, it suffices to find a set X
such that $a \in X$ and $\forall b \in X, b \in F(X)$.

Inductive types, coinductive types

Consider an algebraic type declaration, such as

```
type nat = 0 | S of nat
```

The $F(X)$ operator is defined as “the values we can construct by applying a constructor once, taking arguments of the algebraic type from X ”.

In our example:

$$F(X) = \{0\} \cup \{S(x) \mid x \in X\}$$

This operator is increasing if all occurrences of the algebraic type in the types of constructors are strictly positive.

The inductive type defined by this declaration is μF , the least fixed point of operator F .

The coinductive type defined by this declaration is νF , the greatest fixed point of operator F .

The inductive interpretation

Inductive nat: Type := 0: nat | S: nat -> nat

$$F(X) = \{0\} \cup \{S(x) \mid x \in X\}$$

nat is the least fixed point of F .

It is also the limit of the sequence $\emptyset, F(\emptyset) = \{0\}, F(F(\emptyset)) = \{0, S(0)\}, \dots$

More generally, nat is the set of terms $X \stackrel{\text{def}}{=} \{S^n(0) \mid n \in \mathbb{N}\}$.

- For every n , $S^n(0) \in F^{n+1}(\emptyset) \subseteq \mu F$.
- Conversely, the set X is stable by F , since $F(X) = \{0\} \cup \{S(S^n(0)) \mid n \in \mathbb{N}\} = X$, hence, by the induction principle, $\mu F \subseteq X$.

The coinductive interpretation

CoInductive conat: Type := 0: conat | S: conat -> conat

$$F(X) = \{0\} \cup \{S(x) \mid x \in X\}$$

conat is the greatest fixed point of F .

It is also the limit of the sequence $U, F(U) = \{0\} \cup \{S(x) \mid x \in U\}, \dots, F^n(U) = \{0, S(0), \dots, S^{n-1}(0)\} \cup \{S^n(x) \mid x \in U\}, \dots$

Every $S^n(0)$ is in conat, since $\mu F \subseteq \nu F$.

However, conat also contained the infinite term ω defined by $\omega = S \omega$.

That's because $\{\omega\} \subseteq F(\{\omega\}) = \{0, \omega\}$, hence, by the coinduction principle, $\{\omega\} \subseteq \nu F$.

Inductive predicates, coinductive predicates

The same approach extends to inductive families, and in particular to predicates defined by axioms and inference rules.

Example: the predicate “being even” over natural numbers.

$$\text{even}(0) \qquad \frac{\text{even}(n)}{\text{even}(S(S(n)))}$$

This set of rules corresponds to an operator $F(X)$ that computes the facts we can deduce from the facts X by applying one rule:

$$F(X) = \{\text{even}(0)\} \cup \{\text{even}(S(S(n))) \mid \text{even}(n) \in X\}$$

The inductive interpretation

```
Inductive even: conat -> Prop :=  
  | even_0: even 0  
  | even_S: forall n, even n -> even (S(S n)).
```

$$F(X) = \{\text{even}(0)\} \cup \{\text{even}(S(S(n))) \mid \text{even}(n) \in X\}$$

even is defined as μF , the least fixed point F .

Induction principle: to show $\forall n, \text{even}(n) \Rightarrow P(n)$,
it suffices to show $P(0)$ and $\forall n, P(n) \Rightarrow P(S(S(n)))$.

We can characterize even: (with $f : \text{nat} \rightarrow \text{conat}$ canonical injection)

- $\forall n : \text{nat}, \text{even}(f(n + n))$ by induction over $n : \text{nat}$
- $\forall m, \text{even}(m) \Rightarrow \exists n, m = f(n + n)$ by induction over $\text{even}(m)$
- $\forall m, \text{even}(m) \Rightarrow m \neq \omega$ by induction over $\text{even}(m)$

The coinductive interpretation

```
CoInductive coeven: conat -> Prop :=  
  | coeven_0: coeven 0  
  | coeven_S: forall n, coeven n -> coeven (S(S n)).
```

$$F(X) = \{\text{coeven}(0)\} \cup \{\text{coeven}(S(S(n))) \mid \text{coeven}(n) \in X\}$$

coeven is defined as νF , the greatest fixed point of F .

Coinduction principle: to show $\forall n, P(n) \Rightarrow \text{coeven}(n)$,
it suffices to show $\forall n, P(n) \Rightarrow n = 0 \vee \exists m, n = S(S(m)) \wedge P(m)$.

We can therefore show:

- $\forall n, \text{even}(n) \Rightarrow \text{coeven}(n)$ by coinduction, taking $P(n) = \text{even}(n)$.
- $\text{coeven}(\omega)$ by coinduction, taking $P(n) = (n = \omega)$.

||

Infinite trees, infinite derivations

An interpretation in terms of trees


A value of a (co-)inductive type can be viewed as a tree, with
at leaves: constant constructors,
at nodes: constructors of arity n , where n = number of sub-trees.

(We treat C of $\text{nat} \rightarrow t$ as a constructor with infinite arity.)

Values of a co-inductive type are finite or infinite trees.

Values of an inductive type are trees where all branches are finite.

Values of an inductive type whose constructors have finite arities are finite trees.



An interpretation in terms of derivations

Likewise, a proof term for a (co-)inductive predicate can be viewed as a derivation tree, with at leaves: instances of axioms, at nodes: instances of inference rules with n premises, n = number of sub-trees.

A coinductive predicate holds iff it is the conclusion of a finite or infinite derivation.

An inductive predicate holds iff it is the conclusion of a derivation where all branches are finite.

An inductive predicate where all rules have a finite number of premises holds iff it is the conclusion of a finite derivation.

$$\frac{\text{even}(0)}{\text{even}(S(S(0)))} \qquad \frac{\frac{\frac{\vdots}{\text{coeven}(\omega)}}{\text{coeven}(\omega)}}{\text{coeven}(\omega)}$$

Structural recursion, productive corecursion

A recursive function $f : ind \rightarrow t$ over an inductive type ind must be **structural**: recursive calls $f(y)$ are done over strict sub-terms y of the argument.

```
Fixpoint f(n: nat) : t :=  
  match n with  
  | 0 => ...  
  | S p => ... f p ✓ ... f n ✗ ... end
```

Consequently: the computation $f(n)$ always terminates.

Structural recursion, productive corecursion

A recursive function $f : ind \rightarrow t$ over an inductive type ind must be **structural**: recursive calls $f(y)$ are done over strict sub-terms y of the argument.

A corecursive function $f : t \rightarrow coind$ returning a coinductive type must be **productive**: recursive calls $f(y)$ must be strict sub-terms of the result.

```
CoFixpoint f(x: t) : conat :=
  match x with
  | ... => 0 ✓
  | ... => S(f(...)) ✓
  | ... => f(...) ✗
  | ... => g(f(...)) ✗
```

Consequently: the computation of the head constructor of $f(x)$ always terminates.

Examples of corecursive definitions

The number “positive infinity”:

```
CoFixpoint omega : conat := S omega. ✓
```

The sum of two natural numbers, possibly infinite:

```
CoFixpoint add (p q: conat) : conat :=  
  match p with  
  | 0 => q ✓ (* no recursive call *)  
  | S p' => S (add p' q) ✓ (* recursive call guarded by S *)
```

Subtraction cannot be defined: `sub omega omega` would diverge while determining the head constructor.

```
CoFixpoint sub (p q: conat) : conat :=  
  match p, q with  
  | 0, _ => 0 ✓ (* no recursive call *)  
  | _, 0 => p ✓ (* no recursive call *)  
  | S p', S q' => sub p' q' ✗ (* unguarded recursive call *)  
  end.
```

Examples of coinductive proofs

In the spirit of Curry-Howard, proofs by coinduction are corecursive definitions of proof terms.

For instance, let's define the coinductive predicate `infinite` as

```
CoInductive infinite: conat -> Prop :=
  | infinite_S: forall n, infinite n -> infinite (S n).
```

Then, the proof of `infinite omega` is, morally,

```
CoFixpoint omega_infinite: infinite omega :=
  infinite_S omega omega_infinite.
```

This is not quite right because `omega` is not convertible with `S omega` (see next slide), hence we must explicitly apply the (provable) equality

`omega_eq: omega = S omega`.

```
CoFixpoint omega_infinite: infinite omega :=
  eq_ind_r _ (infinite_S omega omega_infinite) omega_eq.
```

Reduction rules

Just like recursive definitions (`Fixpoint`), corecursive definitions (`CoFixpoint`) cannot be arbitrarily “unrolled”, since this would cause nontermination:

$$\text{omega} \rightarrow S(\text{omega}) \rightarrow S(S(\text{omega})) \rightarrow \dots \quad \times$$

The reduction rule used by Coq: a corecursive definition $\nu x. a$ expands to $a\{x \leftarrow \nu x. a\}$ only when it is argument of a match:

$$\text{match } \nu x. a \text{ with } \dots \rightarrow \text{match } a\{x \leftarrow \nu x. a\} \text{ with } \dots$$

Compare with the rule for recursive definitions $\mu f. \lambda x. a$, which expand only when applied to a constructor:

$$(\mu f. \lambda x. a) (C b) \rightarrow a\{f \leftarrow \mu f. \lambda x. a, x \leftarrow C b\}$$

Unrolling equalities

Even if it does not hold by conversion, we can show the equality $\text{omega} = S \text{ omega}$ by proof.

First we show an extensionality property for values of type `conat`:

```
Lemma unroll: forall (n: conat),  
  n = match n with 0 => 0 | S m => S m end.
```

Then, `unroll omega` has the following type, which reduces:

```
omega = match omega with 0 => 0 | S m => S m end  
→ omega = match S omega with 0 => 0 | S m => S m end  
→ omega = S omega
```

Typing is not preserved by reductions

(E. Giménez, *Un Calcul de Constructions Infinies et son application à la vérification de systèmes communicants*, PhD thesis, ENS Lyon, 1996)

We have the following reductions and typings:

$$\begin{array}{l} \text{unroll } \omega : \omega = S \omega \\ * \downarrow \qquad \qquad \qquad \updownarrow \text{ (not convertible)} \\ \text{eq_refl } \omega : \omega = \omega \end{array}$$

This invalidates preservation of typing by reduction (*subject reduction*):

if $M : t$ and $M \rightarrow M'$ then $M' : t$.

This doesn't mean that Coq + coinductive types is inconsistent; only that a consistency proof is more difficult. (See Giménez's PhD.)

III

Coinductive types, general recursion,
and the partiality monad

Partial computations

(V. Capretta, *General recursion via coinductive types*, LMCS(1), 2005)

```
CoInductive delay (A: Type) : Type :=  
  | now: A -> delay A  
  | later: delay A -> delay A.
```

`delay A` represents computations that return a value of type `A` if they terminate.

The `later` constructor materializes one step of computation.

The type `delay` being coinductive, we can have infinitely many computation steps, and therefore a computation that does not terminate.

Example:

```
CoFixpoint bottom (A: Type) : delay A := later (bottom A).
```

Partial computations

```
CoInductive delay (A: Type) : Type :=  
  | now: A -> delay A  
  | later: delay A -> delay A.
```

We characterize inductively the terminating computations, coinductively the diverging computations:

```
Inductive terminates (A: Type) : delay A -> A -> Prop :=  
  | terminates_now:  
    forall v, terminates (now v) v  
  | terminates_later:  
    forall a v, terminates a v -> terminates (later a) v.
```

```
CoInductive diverges (A: Type) : delay A -> Prop :=  
  | diverges_later:  
    forall a, diverges a -> diverges (later a).
```

General recursion

We can define arbitrary general recursive functions with a delay result type, provided all recursive calls are guarded by a later constructor.

```
CoFixpoint syracuse (n: nat): delay unit :=  
  if Nat.eqb n 1 then now tt  
  else if Nat.even n then later (syracuse (Nat.div n 2))  
  else later (syracuse (3 * n + 1)).
```

We can, then, reason over termination or divergence of the function.

```
Conjecture Collatz_1:  
  forall n, n >= 1 -> terminates (syracuse n) tt.  
Conjecture Collatz_2:  
  exists n, n >= 1  $\wedge$  diverges (syracuse n).
```

A fixed-point combinator

Consider $F : (A \rightarrow \text{delay } B) \rightarrow (A \rightarrow \text{delay } B)$. We can construct a function $Y F : A \rightarrow \text{delay } B$ by iterating F from $\lambda x. \text{bottom}$ and taking “the first defined result”:

$$\begin{aligned} & (\lambda x. \text{bottom}) a \quad \cdot \quad \cdot \quad \cdot \quad \cdot \quad \cdot \quad \cdot \quad \cdot \quad \cdot \quad \cdot \\ & F(\lambda x. \text{bottom}) a \quad \cdot \quad \cdot \quad \cdot \quad \cdot \quad \cdot \quad \cdot \quad \cdot \quad \cdot \quad v \\ & F^2(\lambda x. \text{bottom}) a \quad \cdot \quad \cdot \quad \cdot \quad \cdot \quad v \\ & F^3(\lambda x. \text{bottom}) a \quad \cdot \quad \cdot \quad \cdot \quad \cdot \quad v \end{aligned}$$

Under some hypotheses over F , the fixed-point equation holds, namely the equivalence between $Y F a$ and $F (Y F) a$.

The partiality monad

The `delay` type is a monad, with the `now` constructor as `ret` operation, and the `bind` operation defined as sequencing of two computations.

```
CoFixpoint bind (A B: Type)
              (a: delay A) (f: A -> delay B) : delay B :=
  match a with
  | now v => later (f v)
  | later a' => later (bind a' f)
  end.
```

The expected properties of sequencing hold, for instance `bind a f` diverges iff `a` diverges or `a` terminates with `v` and `f v` diverges.

Productivity issues

Owing to the syntactic productivity criterion, the `bind` we just defined is often unusable inside a coinductive definition.

Example: McCarthy's 91 function.

$$M(n) = \begin{cases} n - 10 & \text{if } n > 100 \\ M(M(n + 11)) & \text{if } n \leq 100 \end{cases}$$

```
CoFixpoint M (n: nat) : delay nat :=  
  if Nat.leb n 100  
  then bind (M (n + 11)) (fun x => M x)  
  else now (n - 10).
```

The corecursive calls `M (n + 11)` and `M x` are rejected because they occur under the `bind`, which is not a constructor of type `delay`.

Going through the free monad

(N. A. Danielsson, *Beating the Productivity Checker Using Embedded Languages*, 2010)

We can work around the issue by presenting the monad as a coinductive type, whose constructors are the operations of the partiality monad: `ret`, `bind`, and `latr`.

```
CoInductive mon: Type -> Type :=
  | ret: forall (A: Type), A -> mon A
  | bind: forall (A B: Type), mon A -> (A -> mon B) -> mon B
  | latr: forall (A: Type), mon A -> mon A.
```

Going through the free monad

Now, function 91 is productive, indeed:

```
CoFixpoint M (n: nat) : mon nat :=  
  if Nat.leb n 100  
  then bind (M (n + 11)) (fun x => M x)  
  else ret (n - 10).
```

Going through the free monad

The description of a computation, of type `mon A`, is turned into a computation, of type `delay A`, by the following function:

```
CoFixpoint interp (A: Type) (m: mon A) : delay A :=
  match m with
  | ret v => now v
  | latr m => later (interp m)
  | bind (ret v) f => later (interp (f v))
  | bind (latr m) f => later (interp (bind m f))
  | bind (bind a f) g =>
      later (interp (bind a (fun v => bind (f v) g)))
  end.
```

Note the use of the monadic laws `bind-ret` and `bind-bind` so as to reduce every `bind` to `later(interp a)`

IV

Codata and copatterns

Types defined by observations

Inductive types are finitely generated by their constructors. Likewise, we can say that **coinductive types are finitely determined by their projections**, that is, by the **observations** we can make over the values of those types.

Example: streams have two projections,

$$\text{hd} : \text{stream } A \rightarrow A \quad \text{tl} : \text{stream } A \rightarrow \text{stream } A$$

A stream s is entirely determined by the results of the observations $\text{hd}(s), \text{hd}(\text{tl}(s)), \dots, \text{hd}(\text{tl}^n(s)), \dots$

Remark: a function $f : A \rightarrow B$ is also a “black box” determined by the results of the applications (observations) $f a_1, \dots, f a_n, \dots$

Data and codata

This suggests to classify types into

- Data types: integers, Booleans, $A \times B$, $A + B$, all inductive types.
- Codata types: functions $A \rightarrow B$, streams, all coinductive types.

Inductive types are presented as sums labeled by constructors.
Likewise, coinductive types are presented as products labeled by projections (i.e. as records).

```
list A := ⟨ nil | cons of A × list A ⟩  
stream A := { hd: A; tl: stream A }
```

Defining codata

Codata is used by applying projections. But how is codata defined? By equations that define how it reacts to projections!

Examples: the stream `from n` of integers $n, n + 1, n + 2, \dots$:

```
(from n).hd = n  
(from n).tl = from (n + 1)
```

Mapping a function f over every element of a stream:

```
(map f s).hd = f (s.hd)  
(map f s).tl = map f (s.tl)
```


Definitions by equations and pattern-matching

This style of definition is also present in Haskell and Agda to define recursive functions over inductive data, such as

```
double 0 = 0
double (S x) = S (S (double x))
```

Symmetry: in the case of data, we need one equation per possible constructor of the argument; in the case of codata, we need one equation per possible projection of the result.

```
(map f s).hd = f (s.hd)           map f nil = nil
(map f s).tl = map f (s.tl)       map f (cons h t) =
                                   cons (f h) (map f t)
```

Patterns and copatterns

(Abel, Pientka, Thibodeau, Setzer, *Copatterns*, POPL 2013)

Abel et al propose to unify these two styles of definitions by equations with the help of **patterns** and **copatterns**.

Patterns (= possible shapes for data)

$p ::= x$	variable
$Cstr\ p_1 \cdots p_n$	constructor

Copattern (= possible observations over codata)

$q ::= \square$	the object being defined
$q\ p$	function application
$q.\text{Proj}$	projection

A piece of codata is, then, a list of (copattern, expression) cases.

For instance, $\lambda x.a$ is $[(\square x, a)]$

and $\{\text{hd} = a, \text{tl} = b\}$ is $[(\square.\text{hd}, a); (\square.\text{tl}, b)]$.

Examples of nested copatterns

Fibonacci numbers, inductively:

```
fib 0          = S 0
fib (S 0)      = S 0
fib (S (S x)) = fib x + fib (S x)
```

Fibonacci numbers, as a stream:

```
fibs .hd      = S 0
fibs .tl .hd  = S 0
fibs .tl .tl  = zipWith plus fibs (fibs.tl)
```

The stream $n, n - 1, \dots, 1, 0, N, N - 1, \dots, 0, N, \dots$:

```
(cycle n) .hd      = n
(cycle 0) .tl      = cycle N
(cycle (S n)) .tl  = cycle n
```

Reduction rules

Each equation is interpreted as a reduction rule (from left to right):

$$\begin{aligned} \text{cycle (S(S(S 0))).t1.t1.hd} &\rightarrow \text{cycle (S(S 0)).t1.hd} \\ &\rightarrow \text{cycle (S 0).hd} \\ &\rightarrow \text{S 0} \end{aligned}$$

The issue in Coq (non-preservation of typing) is avoided because there are no equalities between a piece of codata and its expansion.

For instance, s and $\{\text{hd} = s.\text{hd}; \text{t1} = s.\text{t1}\}$ react identically to projections, but have no reasons to be equal.

Productivity and structural recursion

To guarantee normalization, simple syntactic conditions seem sufficient:

- Productivity: every copattern must start with a projection
- Structural recursion: if we make a recursive call over a variable x , it must appear under a constructor in the copattern.

$F\ x = F\ (S\ x)$ ✗ $(F\ x).\text{tl} = F\ (S\ x)$ ✓ $F\ (S\ x) = G(F\ x)$ ✓

There are complex copatterns where productivity is less obvious:

```
fibs .tl .tl = zipWith plus fibs (fibs.tl)
```

Agda uses sized types for finer control of productivity.

(Abel and Pientka, *Wellfounded Recursion with Copatterns and Sized Types*, JFP(26), 2016.)

V

Reactive programming and guarded recursion

Reactive programming

Broadly speaking: any program whose purpose is to react via computations to events coming from the outside world.

Example: a spreadsheet in “automatic recalculate” mode.

In this lecture: any program that receives a sequence of input values i_0, i_1, i_2, \dots and computes incrementally and “at the same pace” a sequence of output values o_0, o_1, o_2, \dots

Reactive programming = programming with streams?

It is convenient to use streams to represent the sequences of inputs, of outputs, or of intermediate results.

However, a reactive program is not any function from streams to streams. The function must be **causal**: the n -th output o_n depends on the past inputs i_0, \dots, i_n only, not on the future inputs e_{n+1}, e_{n+2}, \dots

Examples of causal functions (✓) and non-causal functions (✗):

✓ $F\ s = \text{map } f\ s$

✓ $F\ s = \text{cons } 0\ s$

✗ $F\ s = \text{tl } s$

✗ $F\ s = \text{cons } (\text{hd } s) (F (\text{tl } (\text{tl } s)))$

✓ $F\ s = \text{cons } (\text{hd } s) (\text{cons } (\text{hd } s) (F (\text{tl } (\text{tl } s))))$

Lustre, a causal-by-construction language

A Lustre program defines a set of streams X_1, X_2, \dots by a set of mutually-recursive equations $X_i = E_i$.

The expression language E_i is sufficiently restricted to guarantee causality. It includes:

- Pointwise computations: $X + Y$ is `zipWith (+) X Y`.
- Temporal operators that reduce to `cons` over streams:

$$\begin{aligned} \text{cst fby } X &= \text{cons cst } X \\ \text{pre } X &= \text{cons nil } X \\ X \rightarrow Y &= \text{cons (hd } X) Y \end{aligned}$$

- Sampling operators over sub-clocks (represented as Boolean streams).

$$\begin{aligned} X \text{ when } C &= \text{cons (if hd } C \text{ then Pre(hd } X) \text{ else Abs)} \\ &\quad ((\text{tl } X) \text{ when } (\text{tl } C)) \end{aligned}$$

Enforcing causality by typing

For additional flexibility (higher-order functions), we can replace syntactic causality conditions by type constraints.

Krishnaswami and Benton (2011–2013) extend types with a “later” modality, written $\triangleright A$.

Intuitively, values of type $\triangleright A$ are values of type A that must not be used immediately (causality would be violated), but can be used at the next time step.

$$\text{hd} : \text{stream } A \rightarrow A$$
$$\text{tl} : \text{stream } A \rightarrow \triangleright(\text{stream } A)$$
$$\text{cons} : A \rightarrow \triangleright(\text{stream } A) \rightarrow \text{stream } A$$

Examples of typings

(In the language of Clouston et al, *The guarded lambda-calculus*, LMCS(12), 2016)

With $\text{fix} : (\triangleright A \rightarrow A) \rightarrow A$, $\text{next} : A \rightarrow \triangleright A$, and

$\otimes : \triangleright(A \rightarrow B) \rightarrow \triangleright A \rightarrow \triangleright B$

$\text{map} = \lambda f. \text{fix}(\lambda m. \lambda s. \text{cons } (f \text{ (hd } s)) \ (m \otimes \ (\text{tl } s)))$

$: (A \rightarrow B) \rightarrow \text{stream } A \rightarrow \text{stream } B$

with $m : \triangleright(\text{stream } A \rightarrow \text{stream } B)$

$\text{nats} = \text{fix}(\lambda s. \text{cons } 0 \ (\text{next } (\text{map } S) \ \otimes \ s))$

$: \text{stream } \text{nat}$

with $s : \triangleright(\text{stream } \text{nat})$

Typing guarantees causality and productivity, while the definition

$\text{CoFixpoint } \text{nats} := \text{cons } 0 \ (\text{map } S \ \text{nats})$

doesn't even pass Coq's syntactic productivity check.

Guarded recursive types

(Nakano, *A modality for recursion*, LICS 2000.)

We can form the recursive type $\mu\alpha. \tau$ provided α is **guarded** in τ :
all occurrences of α are “under” a \triangleright modality.

Examples:

$\mu\alpha. \mathbf{A} \times \triangleright\alpha$	streams (infinite lists)
$\mu\alpha. \mathbf{unit} + \mathbf{A} \times \triangleright\alpha$	finite or infinite lists
$\mu\alpha. \mathbf{bool} \times \triangleright\alpha \times \triangleright\alpha$	infinite binary trees (= deterministic automata)
$\mu\alpha. \mathbf{unit} + \alpha$ ❌	not guarded

Guarded recursive types

(Clouston et al, *The guarded lambda-calculus*, LMCS(12), 2016)

In the “topos of trees” (lectures of Jan 9th 2019), these guarded recursive types are interpreted by time-indexed families of non-recursive types (hence not coinductive).

Example: the type of streams $\mu\alpha. A \times \triangleright\alpha$.

at time 0: `unit`

at time 1: `A × unit`

at time 2: `A × A × unit`

⋮

at time n : `A × ⋯ × A × unit` (lists of length n)

Example: $\mu\alpha. \text{unit} + A \times \triangleright\alpha$ at time $n =$ lists of length $\leq n$.

From guarded recursive types to coinductive types

We recover the usual coinductive types, freely usable at any time, via another modality: \Box , “forever”.

Semantically, $\Box A$ is the “constant object” that is the limit of the interpretations of A at time n when $n \rightarrow \infty$. For instance:

$$\begin{aligned}\Box(\mu\alpha. A \times \triangleright\alpha) &\approx \lim_{n \rightarrow \infty} (\text{lists of } A \text{ of length } n) \\ &\approx \text{infinite lists of } A\end{aligned}$$

$$\begin{aligned}\Box(\mu\alpha. \text{unit} + A \times \triangleright\alpha) &\approx \lim_{n \rightarrow \infty} (\text{lists of } A \text{ of length } \leq n) \\ &\approx \text{finite or infinite lists of } A\end{aligned}$$

Restriction: A must be closed in $\Box A$.

(For example, this prohibits $\mu\alpha. \dots \Box \triangleright \alpha$)

From guarded recursive types to coinductive types

The \Box modality behaves globally like the “necessarily” modality in S4 intuitionistic logic.

$$\frac{\Gamma \vdash a : \Box A}{\Gamma \vdash \text{unbox } a : A} \qquad \frac{\vec{x} : \vec{C} \vdash a : A}{\vec{x} : \vec{C} \vdash \text{box}(a) : \Box A} \qquad \frac{\vec{x} : \vec{C} \vdash a : \triangleright A}{\vec{x} : \vec{C} \vdash \text{prev}(a) : A}$$

C stands for a type constant over time: $C ::= \text{nat} \mid C \rightarrow C \mid \Box A$.

The `box` and `unbox` operators make it possible to define non-causal functions over coinductive types, such as

$$\begin{aligned} \text{hd}_C : \Box(\text{stream } A) \rightarrow A &= \lambda s. \text{hd}(\text{unbox}(s)) \\ \text{tl}_C : \Box(\text{stream } A) \rightarrow \Box(\text{stream } A) &= \lambda s. \text{box}(\text{prev}(\text{tl}(\text{unbox}(s)))) \end{aligned}$$

From guarded recursion to corecursion

There is no fixed point operator corresponding to corecursion. Instead, we must write guarded recursions, then apply `box`, thus guaranteeing productivity.

Example: the stream function “take every other element”. We first define the function that produces a guarded recursive stream

$$\begin{aligned} F &: \square(\text{stream } A) \rightarrow \text{stream } A \\ &= \text{fix}(\lambda f. \lambda s. \text{cons}(\text{hd}_c(s))(g \circledast \text{next}(\text{tl}_c(\text{tl}_c(s)))))) \end{aligned}$$

then, using `box`, the function that produces a coinductive stream:

$$\begin{aligned} G &: \square(\text{stream } A) \rightarrow \square(\text{stream } A) \\ &= \lambda s. \text{box}(F(s)) \end{aligned}$$

VI

Concluding remarks

Happy like Sisyphus?

Using these techniques for coinductive definitions and proofs, we can “look infinity square in the eye” and program and reason directly with infinite objects.

All this also works in type theory and in systems such as Agda and Coq.

However, productivity conditions are very strict, and coinductive reasoning remains difficult.

⇒ more flexible proof principles, such as parameterized coinduction (PACO, C. K. Hur et al, 2013) and coinduction “all the way up” (D. Pous, 2016).

As shown by the example of reactive programming, “finitary” approaches (step-indexing, finite approximations) are sometimes preferable to coinductive approaches.

VII

Further reading

Further reading

A tutorial on set-theoretic coinduction, with application to subtyping:

- B. C. Pierce: *Types and Programming Languages*, chapter 21, MIT Press, 2002.

Coinduction in Coq:

- Y. Bertot and P. Castéran: *Interactive Theorem Proving and Program Development*, chapter 14.

The codata approach:

- A. Abel, B. Pientka, D. Thibodeau, A. Setzer: *Copatterns: programming infinite structures by observations*. POPL 2013. <http://www.cse.chalmers.se/~abela/pop113.pdf>

The guarded recursion approach:

- R. Clouston, A. Bizjak, H. B. Grathwohl, L. Birkedal: *The Guarded Lambda-Calculus: Programming and Reasoning with Guarded Recursion for Coinductive Types*. LMCS 12(3) 2016, <https://arxiv.org/abs/1606.09455>